NAG Library Routine Document

G01JDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01JDF calculates the lower tail probability for a linear combination of (central) χ^2 variables.

2 Specification

```
SUBROUTINE G01JDF (METHOD, N, RLAM, D, C, PROB, WORK, IFAIL)
INTEGER              N, IFAIL
REAL (KIND=nag_wp) RLAM(N), D, C, PROB, WORK(N+1)
CHARACTER(1)         METHOD
```

3 Description

Let $u_1, u_2, \ldots, u_n$ be independent Normal variables with mean zero and unit variance, so that $u_1^2, u_2^2, \ldots, u_n^2$ have independent χ^2-distributions with unit degrees of freedom. G01JDF evaluates the probability that

$$\lambda_1 u_1^2 + \lambda_2 u_2^2 + \cdots + \lambda_n u_n^2 < d\left(u_1^2 + u_2^2 + \cdots + u_n^2\right) + c.$$

If $c = 0.0$ this is equivalent to the probability that

$$\frac{\lambda_1 u_1^2 + \lambda_2 u_2^2 + \cdots + \lambda_n u_n^2}{u_1^2 + u_2^2 + \cdots + u_n^2} < d.$$

Alternatively let

$$\lambda_i^* = \lambda_i - d, \quad i = 1, 2, \ldots, n,$$

then G01JDF returns the probability that

$$\lambda_1^* u_1^2 + \lambda_2^* u_2^2 + \cdots + \lambda_n^* u_n^2 < c.$$

Two methods are available. One due to Pan (1964) (see Farebrother (1980)) makes use of series approximations. The other method due to Imhof (1961) reduces the problem to a one-dimensional integral. If $n \geq 6$ then a non-adaptive method described in D01BDF is used to compute the value of the integral otherwise D01AJF is used.

Pan's procedure can only be used if the λ_i^* are sufficiently distinct; G01JDF requires the λ_i^* to be at least 1% distinct; see Section 9. If the λ_i^* are at least 1% distinct and $n \leq 60$, then Pan's procedure is recommended; otherwise Imhof's procedure is recommended.

4 References

Farebrother R W (1980) Algorithm AS 153. Pan's procedure for the tail probabilities of the Durbin–Watson statistic *Appl. Statist.* **29** 224–227

Imhof J P (1961) Computing the distribution of quadratic forms in Normal variables *Biometrika* **48** 419–426

Pan Jie–Jian (1964) Distributions of the noncircular serial correlation coefficients *Shuxue Jinzhan* **7** 328–337

5 Arguments

1: METHOD – CHARACTER(1) *Input*

On entry: indicates whether Pan's, Imhof's or an appropriately selected procedure is to be used.

METHOD = 'P'
: Pan's method is used.

METHOD = 'I'
: Imhof's method is used.

METHOD = 'D'
: Pan's method is used if λ_i^*, for $i = 1, 2, \ldots, n$ are at least 1% distinct and $n \leq 60$; otherwise Imhof's method is used.

Constraint: METHOD = 'P', 'I' or 'D'.

2: N – INTEGER *Input*

On entry: n, the number of independent standard Normal variates, (central χ^2 variates).

Constraint: $\mathrm{N} \geq 1$.

3: RLAM(N) – REAL (KIND=nag_wp) array *Input*

On entry: the weights, λ_i, for $i = 1, 2, \ldots, n$, of the central χ^2 variables.

Constraint: $\mathrm{RLAM}(i) \neq \mathrm{D}$ for at least one i. If METHOD = 'P', then the λ_i^* must be at least 1% distinct; see Section 9, for $i = 1, 2, \ldots, n$.

4: D – REAL (KIND=nag_wp) *Input*

On entry: d, the multiplier of the central χ^2 variables.

Constraint: $\mathrm{D} \geq 0.0$.

5: C – REAL (KIND=nag_wp) *Input*

On entry: c, the value of the constant.

6: PROB – REAL (KIND=nag_wp) *Output*

On exit: the lower tail probability for the linear combination of central χ^2 variables.

7: WORK($\mathrm{N} + 1$) – REAL (KIND=nag_wp) array *Workspace*

8: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N < 1,
or D < 0.0,
or METHOD $\neq$ 'P', 'I' or 'D'.

IFAIL $= 2$

On entry, RLAM$(i) =$ D for all values of i, for $i = 1, 2, \ldots, n$.

IFAIL $= 3$

On entry, METHOD $=$ 'P' yet two successive values of the ordered λ_i^*, for $i = 1, 2, \ldots, n$, were not at least 1% distinct.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

On successful exit at least four decimal places of accuracy should be achieved.

8 Parallelism and Performance

G01JDF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

Pan's procedure can only work if the λ_i^* are sufficiently distinct. G01JDF uses the check $\left|w_j - w_{j-1}\right| \geq 0.01 \times \max\left(\left|w_j\right|, \left|w_{j-1}\right|\right)$, where the w_j are the ordered nonzero values of λ_i^*.

For the situation when all the λ_i are positive G01JCF may be used. If the probabilities required are for the Durbin–Watson test, then the bounds for the probabilities are given by G01EPF.

10 Example

For $n = 10$, the choice of method, values of c and d and the λ_i are input and the probabilities computed and printed.

10.1 Program Text

```
    Program g01jdfe

!     G01JDF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g01jdf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter              :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)              :: c, d, prob
      Integer                         :: ifail, n
      Character (1)                   :: method
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: rlam(:), work(:)
!     .. Executable Statements ..
      Write (nout,*) 'G01JDF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Read in the problem size
      Read (nin,*) n, method, d, c

      Allocate (rlam(n),work(n+1))

!     Read in data
      Read (nin,*) rlam(1:n)

!     Calculate probability
      ifail = 0
      Call g01jdf(method,n,rlam,d,c,prob,work,ifail)

!     Display results
      Write (nout,99999) ' Values of lambda ', rlam(1:n)
      Write (nout,99999) ' Value of D       ', d
      Write (nout,99999) ' value of C       ', c
      Write (nout,*)
      Write (nout,99998) ' Probability = ', prob

99999 Format (1X,A,10F6.2)
99998 Format (1X,A,F10.4)
    End Program g01jdfe
```

10.2 Program Data

```
G01JDF Example Program Data
10 'P' 1.0 0.0
 -9.0 -7.0 -5.0 -3.0 -1.0 2.0 4.0 6.0 8.0 10.0
```

10.3 Program Results

```
G01JDF Example Program Results

 Values of lambda  -9.00 -7.00 -5.00 -3.00 -1.00  2.00  4.00  6.00  8.00 10.00
 Value of D         1.00
 value of C         0.00

 Probability =     0.5749
```